



SPARQL

27 MARS 2023

Benoît Deshayes, [User:Shonagon](#)

Accéder aux données de Wikidata

Principaux modes

- ❖ Accès direct aux données d'un élément

exemple : <https://www.wikidata.org/wiki/Special:EntityData/Q180736.rdf>

- ❖ Les dumps (exports complets de la base de données)

- ❖ API de MediaWiki

- ❖ API de Wikibase

- ❖ **Endpoint SparQL** : <https://query.wikidata.org/>

Page dédiée à l'accès aux données

https://www.wikidata.org/wiki/Wikidata:Data_access/fr

❖ **SPARQL** = *Sparql Protocol and RDF Query Language*

Un **langage** informatique **de requête** (≠ langage de programmation)

Demandes / Réponses

Une syntaxe

rigoureuse

Des briques fonctionnelles

SELECT, DESCRIBE... SERVICE...

Des éléments documentaires formalisés

wdt:p50 → auteur

wd:Q535 → Victor Hugo


...

SparQL est le langage pour les graphes du web de données


Des éléments Wikidata



Q29530
"La Liberté guidant le peuple"@fr



Q1133821
"peinture"@fr



Q33477
"Eugène Delacroix"@fr



Q29647193
"L'Éducation de la Vierge"@fr



Q142
"France"@fr



Q1782606
"musée national Eugène-Delacroix"@fr



Q3044768
"département des peintures du musée du Louvre"@fr




Q1133821
"La Mort de Sardanapale"@fr




Q3119110
"Jeune Orpheline au cimetière"@fr

Des données modélisées en graphes




Des données modélisées en graphes




Des triplets similaires

Triplet




Sujet	Prédicat	Objet
Q29530	P170	Q33477

Triplets similaires



Q29647193 P170 Q33477



Q1133821 P170 Q33477

Rechercher à partir d'un triplet

=> Toute chose créée par Delacroix correspond à un triplet de la forme



=> Recherche de tous les éléments avec un triplet similaire

Lister toutes les choses correspondant au critère :

?Chose "créé par"/P170 "Eugène Delacroix"/Q33477

s'exprime en SparQL :

```

SELECT ?chose WHERE {
  ?chose wdt:P170 wd:Q33477.
}
  
```

Première requête SparQL

```
SELECT ?chose WHERE {  
  ?chose wdt:P170 wd:Q33477.  
}
```

SELECT : instruction

?chose : variable

Le nom de la variable n'importe pas du moment que celui choisi demeure inchangé

WHERE : clauses

Préfixes :

wd: → [<http://www.wikidata.org/entity/>](http://www.wikidata.org/entity/)

wdt: → [<http://www.wikidata.org/prop/direct/>](http://www.wikidata.org/prop/direct/)

Ainsi

wdt:P170 équivaut à [<http://www.wikidata.org/prop/direct/P170>](http://www.wikidata.org/prop/direct/P170)

wd:Q33477 équivaut à [<http://www.wikidata.org/entity/Q33477>](http://www.wikidata.org/entity/Q33477)

La requête dans l'Endpoint :

<http://w.wiki/3hF>

Règles de base de la syntaxe

```
SELECT ?chose WHERE {  
  ?chose wdt:P170 wd:Q33477.  
}
```

- ❖ Accolades et parenthèses **doivent** aller par paires
- ❖ Clauses basées sur des triplets
- ❖ Caractère . pour terminer un triplet
- ❖ Caractère ? pour les variables
- ❖ Les minuscules / majuscules comptent : ?chose ≠ ?Chose

La requête dans l'Endpoint : <http://w.wiki/3hF>

Interroger Wikidata avec SparQL

- **Le point d'accès (Endpoint)**

<https://query.wikidata.org/>

- **Le manuel**

https://www.mediawiki.org/wiki/Wikidata_Query_Service/User_Manual/fr

- **Des exemples de requêtes**

Exemples sur <https://query.wikidata.org/>

Exercice 1

- **Recherche :**

Quels sont les éléments avec Johannes Vermeer en créateur ?

- **Réponse :**

<http://w.wiki/3hJ>

Afficher les libellés

- avec **SERVICE wikibase** (non standard, propre au logiciel de traitement SparQL Blazegraph)

```
SELECT ?element ?elementLabel
WHERE {
    ?element wdt:P170 wd:Q296.
    SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],fr". }
}
```

➤ <https://w.wiki/6Vo4>

La présence du libellé dans la langue indiquée est facultatif

- avec **rdfs:label**

```
SELECT ?element ?elementLabel
WHERE {
    ?element wdt:P170 wd:Q296.
    ?element rdfs:label ?elementLabel .
    FILTER (lang(?elementLabel)="fr") .
}
```

➤ <https://w.wiki/6Vo5>

La présence du libellé dans la langue indiquée est obligatoire

- Copier/coller à l'usage

Exercice 2

- **Recherche :**

Lister les peintures de Goya avec libellés en espagnol

- **Réponse :**

<https://w.wiki/6Vo7>

ou

<https://w.wiki/6Vo9>

- Utilisation du **caractère « ; »** pour ne pas répéter le sujet

```
SELECT ?element ?elementLabel
WHERE {
  ?element wdt:P170 wd:Q5432.
  ?element wdt:P31 wd:Q3305213.
  SERVICE wikibase:label { bd:serviceParam wikibase:language "fr,es,en". }
}
```

équivalent à

```
SELECT ?element ?elementLabel
WHERE {
  ?element wdt:P170 wd:Q5432 ;
           wdt:P31 wd:Q3305213.
  SERVICE wikibase:label { bd:serviceParam wikibase:language "fr,es,en". }
}
```

Avoir plusieurs variables

- Utilisation du **caractère « ; »** pour ne pas répéter le sujet

```
SELECT ?element ?elementLabel ?lieuLabel
WHERE {
  ?element wdt:P170 wd:Q5432;
           wdt:P31 wd:Q3305213;
           wdt:P276 ?lieu.
  SERVICE wikibase:label { bd:serviceParam wikibase:language "fr" }
}
```

➤ <https://w.wiki/6VoF>

Exercice 3

- **Question :**

Ajouter à la liste des peintures de Goya la date de création et un fichier image

- **Réponse :**

<https://w.wiki/6VoH>

Fonction *year/année* et tri

- Avec année de création

```
SELECT ?oeuvre ?oeuvreLabel (year(?DateCrea) as ?AnneeCrea)
```

```
WHERE {
```

```
  ?oeuvre wdt:P170 wd:Q33477 ;
```

```
          wdt:P571 ?DateCrea .
```

```
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],fr". }
```

```
}
```

➤ <https://w.wiki/5Gnp>

- Ordonner les résultats

```
SELECT ?oeuvre ?oeuvreLabel (year(?DateCrea) as ?AnneeCrea)
```

```
WHERE {
```

```
  ?oeuvre wdt:P170 wd:Q33477 ;
```

```
          wdt:P571 ?DateCrea .
```

```
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],fr". }
```

```
} ORDER BY ?AnneeCrea
```

➤ <https://w.wiki/5Gnr>

Œuvres d'Eugène Delacroix avec date de création et en option l'image

#defaultView:ImageGrid

```
SELECT ?oeuvre ?oeuvreLabel (year(?DateCrea) as ?AnneeCrea) ?image
```

```
WHERE {
```

```
  ?oeuvre wdt:P170 wd:Q33477 ;
```

```
    wdt:P571 ?DateCrea .
```

```
OPTIONAL {?oeuvre wdt:P18 ?image .}
```

```
SERVICE wikibase:label { bd:serviceParam wikibase:language "fr" }
```

```
} ORDER BY ?AnneeCrea
```

Exercice 4

- **Question :**

Lister les œuvres de Claude Monet localisées dans des endroits situées en France et afficher :

- l'identifiant Wikidata
- le libellé, préférentiellement en français
- l'année de création
- le lieu
- en option une image

- **Réponse :**

<https://w.wiki/6Voj>

Exercice 5

- **Question :**

A partir de la requête de l'exercice 4, quelle serait la propriété à utiliser pour géolocaliser les œuvres et avec quelle données l'utiliser ?
Compléter la requête.

Une vue par carte peut s'afficher dans la fenêtre de résultat ou ajoutant en tout début de requête :

#defaultView:Map

- **Réponse :**

<https://w.wiki/6Vow>

Instruction DESCRIBE

- **Question :**

Pour obtenir un graphe RDF d'une ressource

DESCRIBE wd:Q180736

➤ <https://query.wikidata.org/#DESCRIBE%20wd%3AQ180736%0A>

Exercice 6

- **Question :**

Lister les œuvres littéraires de Jules Verne avec année de publication

En option : année de publication, image

Tri par année de publication

Avec nature de l'œuvre

[Note 1 : Auteur -> propriété P50

Note 2 : Utiliser la propriété « Date de publication »

Note 3 : construire la requête progressivement]

- **Réponse :**

<https://w.wiki/5Gnu>

(Pour aller plus loin, ajouter l'identifiant catalogue de la BnF : <https://w.wiki/5Gnx>
puis un lien vers le catalogue BnF <https://w.wiki/5Gnz>)

A solid green vertical bar is positioned to the left of the main title.

Aller plus loin avec SparQL

Inclure les sous-classes dans la requête

- Sculptures et sous-classes de sculptures (ex. statues) de Jeanne d'Arc

```
SELECT ?oeuvre ?oeuvreLabel ?image ?coord
```

```
WHERE {
```

```
  ?oeuvre wdt:P31/wdt:P279* wd:Q860861.
```

```
  ?oeuvre wdt:P180 wd:Q7226 .
```

```
  ?oeuvre wdt:P625 ?coord.
```

```
  OPTIONAL { ?oeuvre wdt:P18 ?image. }
```

```
  SERVICE wikibase:label { bd:serviceParam wikibase:language "fr" }
```

```
}
```

➤ <https://w.wiki/5Go3>

- [P279](#) : Propriété « Sous-classe de »

Interroger les qualificatifs

dans la série	Voyages extraordinaires ...
ordre dans la série	3 ...
précédé par	Voyages et aventures du Capitaine Hatteras ...
suivi par	De la Terre à la Lune ...

[Q219457](#)

- Liste dans l'ordre les œuvres de la série des *Voyages extraordinaires* de Jules Verne

```
SELECT ?oeuvre ?oeuvreLabel ?ordre
```

```
WHERE {
```

```
  ?oeuvre p:P179 ?DeclarationSerie.
```

```
  ?DeclarationSerie ps:P179 wd:Q3048935 ;
```

```
    pq:P1545 ?ordre.
```

```
  SERVICE wikibase:label { bd:serviceParam wikibase:language "fr" }
```

```
} ORDER BY xsd:integer(?ordre)
```

➤ <http://w.wiki/3hV>

Regroupement

Exemple : éléments iconographiques dans les œuvres de Vermeer

➤ <http://w.wiki/3hc> (213 résultats)

- Faire des regroupements dans une seule ligne de résultats

```
SELECT ?item ?itemLabel ?image (GROUP_CONCAT(DISTINCT(?depeintlb); separator=", ") as ?icono)
```

```
WHERE {
```

```
  ?item wdt:P170 wd:Q41264 .
```

```
  OPTIONAL { ?item wdt:P18 ?image }
```

```
  OPTIONAL { ?item wdt:P180 ?depeint.
```

```
    ?depeint rdfs:label ?depeintlb .
```

```
    filter (lang(?depeintlb) = "fr") .}
```

```
  SERVICE wikibase:label { bd:serviceParam wikibase:language "fr" . }
```

```
} GROUP BY ?item ?itemLabel ?image
```

➤ <http://w.wiki/3hd> (39 résultats)

limiter les résultats

- Personnes nées à Londres

```
SELECT DISTINCT ?item ?Nom
WHERE {
  ?item wdt:P19 wd:Q84 ;
        rdfs:label ?Nom filter (lang(?Nom) = "fr").
} LIMIT 10
```

➤ <https://w.wiki/5Go6>

❖ *Exemple de requêtes avec l'Endpoint SparQL de Wikidata*

<https://query.wikidata.org>

❖ **Wikidata:Tutoriel SPARQL**

https://www.wikidata.org/wiki/Wikidata:SPARQL_tutorial/fr

❖ **Faisons des étincelles avec SparQL**

<https://www.twitch.tv/belett/videos>